Panel Data Methods using Stata
4A: Panels - Unit Root Tests

© A. Colin Cameron
Univ. of Calif. Davis

November 2013

Introduction

- Test whether y_{it} is nonstationary, specifically a unit root
  - Does $y_{it} = y_{i,t-1} + \varepsilon_{it}$?
- In pure time series case, if unit root then test statistics are not normal
  - instead functions of Wiener processes / Brownian Motion
  - and have low power: cannot easily distinguish between $y_{it} = 0.9y_{i,t-1} + \varepsilon_{it}$ and $y_{it} = y_{i,t-1} + \varepsilon_{it}$.
- Panel data may give more power
  - again nonstandard asymptotic theory + intercept(?) + trend(?)
  - in particular, may need to recenter and rescale test statistic
  - but then usually normally distributed
- But - there are many different tests of unit roots in panels
  - $N \to \infty$ or $T \to \infty$ or both; heterogeneity; cross-section correlation
  - for panel case there is no general clear best test
  - good idea to read the original paper before using a test
  - and rely on economic theory to give likely model for y_{it}.

Outline

2. Time Series Unit Root Tests Consequences

- y_t has a unit root if need to first difference to get stationary process
  - examples: random walk with drift and random walk without drift.
- Random walk without drift
  - If $y_t = y_{t-1} + \varepsilon_t$ then $y_t = (y_{t-2} + \varepsilon_{t-1}) + \varepsilon_t = \cdots$
    so $y_t = y_0 + \sum_{s=0}^{t} \varepsilon_s$
    - ★ shocks are permanent as $\Delta\varepsilon_0 = 1 \Longrightarrow \Delta y_t = 1$
- Compared to AR(1)
  - Versus $y_t = \rho y_{t-1} + \varepsilon_t$, $\rho < 1$, then $y_t = \rho(\rho y_{t-2} + \varepsilon_{t-1}) + \varepsilon_t = \cdots$
    so $y_t = \rho^t y_0 + \sum_{s=0}^{t} \rho^{t-s} \varepsilon_s$
  - shocks disappear as $\Delta\varepsilon_0 = 1 \Longrightarrow \Delta y_t = \rho^t \rightarrow 0$.
- Fundamental result: shocks are permanent if unit root.
- Random walk with drift: $y_t = \alpha + y_{t-1} + \varepsilon_t$
  - implies $y_t = y_0 + \alpha t + \sum_{s=0}^{t} \varepsilon_s$
  - so induces a linear time trend as well as being nonstationary.

Dickey-Fuller Test

- Fuller (1976). Rewrite the model

$$\begin{aligned} y_t &= \rho y_{t-1} + \varepsilon_t \\ \Delta y_t &= (\rho - 1) y_{t-1} + \varepsilon_t \\ &= \phi y_{t-1} + \varepsilon_t . \end{aligned}$$

  - unit root if $\phi = 0$ and stationary if $\phi < 0$.
- Dickey Fuller test: test $H_0 : \phi = 0$ against $H_a : \phi < 0$ in

$$\Delta y_t = \phi y_{t-1} + \varepsilon_t$$

- Obvious approach is to test for $\phi = 0$ in regress y_t on y_{t-1}
  - But under $H_0 : \phi = 0$ test statistic has nonstandard distribution
  - use special tables not t tables
  - Stata command `dfuller`

Simulation

- DGP: $y_t = y_{t-1} + \varepsilon_t$ where ε_t iid $N[0,1]$ and $T = 50$.
- Stata code

```
set seed 10101
program unitroot, rclass
drop _all
set obs 50
generate time = _n
tsset time
generate epsilon = rnormal(0,1)
generate yrwalk = 0
replace yrwalk = L.yrwalk + epsilon if time > 1
regress D.yrwalk L.yrwalk
return scalar b2 =_b[L.yrwalk]
return scalar se2 = _se[L.yrwalk]
return scalar t2 = (_b[L.yrwalk]-0)/_se[L.yrwalk]
end
```

- Do 10,000 simulations and summarize results

```
quietly simulate b2=r(b2) se2=r(se2) t2=r(t2), ///
reps(1000) saving(unitroot, replace) nolegend nodots: unitroot
summarize b2 se2 t2
mean b2 se2 t2
histogram t2, normal title("DF Unit root test Case 2")
centile t2, centile(1 2.5 5 10 90 95 97.5 99)
```

- Centiles for t-statistic: 1 2.5 5 10 90 95 97.5 99
  - N[0,1]: -2.57, -1.96, -1.64, -1.28, 1.28, 1.64, 1.96, 2.57
  - Simulation: -3.56, -3.22, -2.92, -2.60, -0.40, -0.02, 0.30, 0.63
  - DF Tables: -3.58, -3.22, -2.93, -2.60, -0.40, -0.03, 0.29, 0.66
- Definitely not normal or $T(48)$.


DF Unit root test
Density
0
.1
.2
.3
.4
.5
-6
-4
-2
0
2
t-statistic for slope coeff from many samples

Dickey-Fuller Tests (continued)

- Complication 1: There are two test statistics
  - The t-statistic $\widehat{\phi}/se(\widehat{\phi})$
  - The z-statistic $T\widehat{\phi}$
    - ★ we can use the z-statistic in addition to the t statistic because when $\phi = 0$ its distribution does not depend on unknown parameters .
- Complication 2: Under H_0 these statistics have nonstandard distributions - functionals of Wiener processes (Brownian motion)
  - asymptotic results use a functional central limit theorem that does not require ε_t to be normally distributed
  - letting $\Psi = \Psi(r)$ denote a detrended Wiener process
    - ★ $T\widehat{\phi} \xrightarrow{d} \frac{\int \Psi(r)d\Psi(r)}{\int \Psi(r)^2 dr}$ and $\frac{\widehat{\phi}}{se(\widehat{\phi})} \xrightarrow{d} \frac{\int \Psi(r)d\Psi(r)}{\sqrt{\int \Psi(r)^2 dr}}$
  - finite sample results assume ε_t is normally distributed

Dickey-Fuller Tests (continued)

- Complication 3: These distributions change when deterministic terms are added to the regression, so $y_t = \mathbf{d}_t'\gamma + \rho y_{t-1} + \varepsilon_t$ where $\mathbf{d}_t$ is constant plus possibly trend
  - Case 1: $y_t = \rho y_{t-1} + \varepsilon_t$ estimated and $H_0 : y_t = y_{t-1} + \varepsilon_t$
  - Case 2: $y_t = \alpha + \rho y_{t-1} + \varepsilon_t$ estimated and $H_0 : y_t = y_{t-1} + \varepsilon_t$
  - Case 3: $y_t = \alpha + \rho y_{t-1} + \varepsilon_t$ estimated and $H_0 : y_t = \delta + y_{t-1} + \varepsilon_t$
  - Case 4: $y_t = \alpha + \delta t + \rho y_{t-1} + \varepsilon_t$ estimated and $H_0 : y_t = \delta + y_{t-1} + \varepsilon_t$
- Most often use cases 2 and 4
  - case 3 unrealistic as H_0 model $y_t = y_0 + \delta t + (\varepsilon_t + \varepsilon_{t-1} + \cdots + \varepsilon_1)$ has time trend
  - lose power as move from case 1 to case 2 to case 4
- Stata `dfuller` uses the t-statistic $\widehat{\phi}/se(\widehat{\phi})$
  - cases 1-4 are, respectively, options (1) noconstant, (2) the default, (3) drift, (4) trend

Unit Root Tests for Time Series

- Complication 4: ε_t is serially correlated
- Dickey and Fuller (1979) - Augmented Dickey-Fuller (ADF) test
  - easiest to understand
  - add lags until ε_t is serially uncorrelated and use original DF tables
  - so estimate $\Delta y_t = \mathbf{d}_t'\gamma + \phi y_{t-1} + \sum_{k=1}^{K} \gamma_k \Delta y_{t-k} + \varepsilon_t$.
  - t-statistic is $\widehat{\phi}/se(\widehat{\phi})$ and z-statistic is $T\widehat{\phi}/(1-\widehat{\gamma}_1 - \cdots - \widehat{\gamma}_T)$
  - determine # lags using AIC or BIC or by specifying K, doing regular t-test on $\widehat{\gamma}_K$, then $\widehat{\gamma}_{K-1}$, then, ...
- Phillips and Perron (1988) test
  - correct the original (unaugmented) DF test for ε_t serially correlated
  - then use the original DF tables
- KPSS (1992) test of $H_0 : \phi < 0$ against $H_a : \phi = 0$
  - OLS regress $y_t = \mathbf{d}_t'\gamma + u_t$, $\widehat{u}_t = y_t - \mathbf{d}_t'\widehat{\gamma}$, $S_t = \sum_{s=1}^{t} \widehat{u}_s^2$
  - $LM = \sum_{t=1}^{T} S_t^2/(T^2 f_0)$; f_0 is estimate of the long-run variance of $\widehat{u}_t$
- Other tests include DF-GLS test coming next.

Unit Root Tests for Time Series (continued)

- Elliott, Rothenberg & Stock (1996) test - DF-GLS test
  - this is viewed as the best unit root test
  - most powerful test against local alternatives $H_0 : \phi = cT$ for fixed c
  - Stata command `dfgls`
- $y_t = \mathbf{d}_t'\gamma + \rho y_{t-1} + \sum_{k=1}^{K} \gamma_k \Delta y_{t-k} + \varepsilon_t$; $d_t = 1$ or $\mathbf{d}_t = (1, t)$
  - DF test jointly estimates γ and ρ and $\gamma_k' s$
  - when $d_t = 1$ equivalently ρ (and $\gamma_k' s$) from OLS of $(y_t - \bar{y})$ on $\rho(y_{t-1} - \overline{y_{t-1}})$ and $\Delta y_{t-k}' s$.
- 1. Instead (GLS step) estimate γ by $\widehat{\gamma}$ from OLS in $(y_t - \bar{\rho} y_{t-1}) = (\mathbf{d}_t - \bar{\rho}\mathbf{d}_t)'\gamma + v_t$
  - where $\bar{\rho} = 1 - 7/T$ for $d_t = 1$ and $\bar{\rho} = 1 - 13.5/T$ for $\mathbf{d}_t = (1, t)$
- 2. ϕ (and $\gamma_k' s$) by OLS in $\Delta y_t^* = \phi y_{t-1}^* + \sum_{k=1}^{K} \gamma_k \Delta y_{t-k}^* + w_t$ where $y_t^* = y_t - \mathbf{d}_t'\widehat{\gamma}$
- 3. Use t-statistic for $\widehat{\phi} = 0$
  - for $d_t = 1$ use DF tables (case 2)
  - for $\mathbf{d}_t = (1, t)$ use tables in Elliott et al.

3. Panel Example: Real Exchange Rates and PPP

- Real Exchange Rate annual data 1970-2003 on 151 countries.
  - Balanced with $T = 34$ and $N = 151$
  - Sometimes restrict to OECD $N = 27$ or G7 $N = 6$
  - USA is the reference country
  - Data from Stata Manual [XT] `xtunitroot`.
- Real exchange rate = nominal exchange rate × (price in home country / price in foreign country)

$$\lambda = EP^*/P$$
$$\ln \lambda = \ln E + \ln P^* - \ln P$$

- Purchasing power parity says **no unit root (instead I(0))**
  - reason: real exchange rate is mean-reverting
  - essentially even if $\ln E$, $\ln P^*$ and $\ln P$ are I(1)
    - ★ they are cointegrated so that $\ln E + \ln P^* - \ln P$ is I(0).
- Also there is no reason to believe there is a trend
  - so unit root tests here do not include a trend.

Data Description

```
. describe

Contains data from pennxrate.dta
  obs:         5,134
 vars:            10                          15 Oct 2012 22:17
 size:       200,226 (99.7% of memory free)   (_dta has notes)
-----------------------------------------------------------------------------------
              storage  display     value
variable name   type   format      label      variable label
-----------------------------------------------------------------------------------
country         str3   %9s
year            int    %8.0g
xrate           float  %9.0g                  Nominal exchange rate
ppp             float  %9.0g                  PWT Purchasing Power Parity index
id              float  %9.0g                  group(country)
capt            float  %9.0g
realxrate       float  %9.0g                  Real exchange rate
lnrxrate        float  %9.0g                  Log real exchange rate
oecd            byte   %8.0g
g7              byte   %8.0g
-----------------------------------------------------------------------------------
Sorted by:  id  year
```

```
. * Summarize data
. sum
```

Variable	Obs	Mean	Std. Dev.	Min	Max
country	0				
year	5134	1986.5	9.811664	1970	2003
xrate	5134	1412.495	36112.62	1.67e-12	1507226
ppp	5134	664.0886	18089.86	1.54e-12	852999.1
id	5134	94.36424	52.87679	1	188
capt	5134	34	0	34	34
realxrate	5134	.632419	.5703317	.005522	11.02636
lnrxrate	5134	-.6375265	.5680524	-5.19902	2.400289
oecd	5134	.1788079	.3832287	0	1
g7	5134	.0397351	.1953552	0	1
lnxrate	5134	1.856842	4.194947	-27.11847	14.22578
lnppp	5134	1.197976	4.097343	-27.19693	13.65651

Panel Summary

- Key variables have both within and between variation
  - lnrxrate is key variable
  - lnppp is ln price ratio and lnxrate is ln nominal exchange rate

```
. xtsum lnrxrate lnxrate lnppp
```

Variable		Mean	Std. Dev.	Min	Max	Observations
lnrxrate	overall	-.6375265	.5680524	-5.19902	2.400289	N = 5134
	between		.4317351	-1.560863	.7197967	n = 151
	within		.370792	-5.137827	1.797904	T = 34
lnxrate	overall	1.856842	4.194947	-27.11847	14.22578	N = 5134
	between		3.451942	-16.00139	7.287732	n = 151
	within		2.399643	-12.46177	23.86295	T = 34
lnppp	overall	1.197976	4.097343	-27.19693	13.65651	N = 5134
	between		3.403155	-16.52128	6.778892	n = 151
	within		2.29809	-13.06217	22.3583	T = 34

Line Charts (for OECD)

- Seem to move together against the reference USA
  - `xtline lnrxrate if oecd == 1, overlay`
  - control for cross-country correlation by including $\bar{y}_t = \frac{1}{T}\sum_{i=1}^{N} y_{it}$.



4. Time Series Unit Root Tests for GBR-USA

- Consider ADF test for a single time series
  - lnrxrate is GBR-USA log real exchange rate

```
. * ADF test with 5 lags
. dfuller lnrxrate if country == "GBR", lags(5)

Augmented Dickey-Fuller test for unit root         Number of obs   =        28

                               ---------- Interpolated Dickey-Fuller ---------
                  Test         1% Critical       5% Critical      10% Critical
               Statistic           Value             Value             Value
------------------------------------------------------------------------------
 Z(t)             -1.632            -3.730            -2.992            -2.626
------------------------------------------------------------------------------
MacKinnon approximate p-value for Z(t) = 0.4666
```

- Do not reject H_0 as $p > .05$
  - conclude there is a unit root!
- But this is very black box.

- The following also shows the associated ADF regression

```
dfuller lnrxrate if country == "GBR", lags(5) regress
```

D.lnrxrate	Coef.	Std. Err.	t	P>|t|	[95% Conf.	Interval]
lnrxrate						
L1.	-.2397465	.1469375	-1.63	0.118	-.5453198	.0658268
LD.	.4322423	.2210537	1.96	0.064	-.0274641	.8919487
L2D.	.0879437	.2176313	0.40	0.690	-.3646454	.5405327
L3D.	-.2028981	.2105659	-0.96	0.346	-.6407938	.2349977
L4D.	-.1348889	.2074825	-0.65	0.523	-.5663723	.2965946
L5D.	.1590723	.1595226	1.00	0.330	-.1726731	.4908177
_cons	-.0083102	.0219679	-0.38	0.709	-.053995	.0373746

- The estimated coefficient $-.239$ is a long way from 0
  - Note, however, that $\widehat{\phi}$ is considerably biased below zero if $\phi = 0$
  - Fail to reject H_0 in part because very noisily estimated
  - And note that H_0 is unit root, not H_0 is no unit root.
- Aside: `dfgls` with 5 lags also does not reject H_0

- The autocorrelations look a lot like AR(1) with $\rho \simeq 0.7$, not like unit root.

```
. * ACF
. corrgram lnrxrate if country == "GBR"

                                          -1       0       1 -1       0       1
 LAG       AC       PAC      Q     Prob>Q  [Autocorrelation]  [Partial Autocor]
-------------------------------------------------------------------------------
1        0.7436   0.7715   20.509  0.0000
2        0.4734  -0.2900    29.08  0.0000
3        0.2276  -0.0478   31.125  0.0000
4        0.0846   0.0984   31.416  0.0000
5        0.0575   0.1024   31.556  0.0000
6        0.0315  -0.1591     31.6  0.0000
7        0.0620   0.1127   31.774  0.0000
8        0.1205   0.1130   32.457  0.0001
9        0.1468   0.0177   33.513  0.0001
10       0.2255   0.4260   36.106  0.0001
11       0.2407   0.0279   39.189  0.0000
12       0.1603  -0.2127   40.619  0.0001
13       0.0141   0.0061    40.63  0.0001
14      -0.1392   0.0853   41.817  0.0001
15      -0.2059   0.1023   44.547  0.0001
```

5. Panel Unit Roots: Overview

- Levin and Lin (1992) seminal paper that introduced many of the ideas
  - need to recenter and rescale DF but ultimately normal
  - results vary with heterogeneity, deterministic trends, ...
- A quite general model for heterogeneous panels is

$$\begin{aligned}\Delta y_{it} &= \phi_i y_{i,t-1} + \alpha_i + \delta_i t + \theta_t + u_{it} \\ &= \phi_i y_{i,t-1} + \mathbf{z}'_{it}\gamma_i + \theta_t + u_{it}\end{aligned}$$

- Three cases of deterministic trends:
  - no constant: $\alpha_i = 0$; $\delta_i = 0$ so $\mathbf{z}'_{it}\gamma_i = 0$
  - constant, no trend: $\delta_i = 0$ so $\mathbf{z}'_{it}\gamma_i = \alpha_i$
  - trend: both nonzero so $\mathbf{z}'_{it}\gamma_i = \alpha_i + \delta_i t$.
- Different tests make different restrictions on this.
  - initially u_{it} i.i.d. (and possibly normal for finite sample results)
  - relax to serially correlated ($\Rightarrow y_{i,t-1}$ endogenous)
  - relax to u_{it} correlated over i (more complicated).

Limit Theory

- Asymptotics may potentially be
  - N fixed $T \to \infty$
  - T fixed $N \to \infty$
  - $T \to \infty$ and $N \to \infty$.
- For $T \to \infty$ and $N \to \infty$ Phillips and Moon (1999 Ecta) consider
  - sequential limits: notation $(T, N \to \infty)_{\text{seq}}$
    - ★ e.g. fix N, let $T \to \infty$, then let $N \to \infty$.
  - diagonal path: notation $T = T(N)$
    - ★ $(T, N) \to \infty$ at rate $T = T(N)$
  - joint limits: notation $N, T \to \infty$
    - ★ simultaneously without any restrictions.

Pedagogical Example

- Levin and Lin (1992)
  - common ρ, heterogeneous intercept, homoskedasticity
  - $y_{it} = \rho y_{i,t-1} + \alpha_i + u_{it}$; $u_{it} \sim \text{iid}[0, \sigma^2]$
  - note: Levin, Lin, Chu (2002) consider more general models
- Demeaning takes care of α_i (Frisch-Waugh Theorem)
  - then $\widehat{\phi} = \widehat{\rho} - 1$ comes from OLS of Δy_{it-1} on $y_{it} - \bar{y}_i$
  - $\sqrt{N} T \widehat{\phi} = \sqrt{N} \frac{\sum_i \frac{1}{T} \sum_t \{(y_{i,t-1} - \bar{y}_i)/\sigma\} \times \Delta\{y_{i,t-1}/\sigma\}}{\sum_{i=1}^N \frac{1}{T^2} \sum_{t=1}^T \{(y_{i,t-1} - \bar{y}_i)/\sigma^2\}}$
  - $\sqrt{N} T \widehat{\phi} = \sqrt{N} \frac{\sum_i a_{iT}}{\sum_i b_{iT}} = \sqrt{N} \frac{a_{NT}}{b_{NT}}$ defining a_{iT}, b_{iT}, a_{NT}, b_{NT}.
- 1. $T \rightarrow \infty$, find a_{iT}, b_{iT} properties using functional CLT's
  - a_{iT} is function of Wiener process and has mean $1/6$ and variance $1/45$
  - b_{iT} is function of Wiener process and has mean -$1/2$ and variance $1/12$
- 2. $N \rightarrow \infty$ and apply CLT's and LLN's to get limit distribution of
  - $\sqrt{N}(T\widehat{\phi} - \frac{\text{E}[a_{NT}]}{\text{E}[b_{NT}]}) = \frac{\sqrt{N}(a_{NT} - \text{E}[a_{NT}])}{b_{NT}} + \text{E}[a_{NT}]\sqrt{N}\left(\frac{1}{b_{NT}} - \frac{1}{\text{E}[b_{NT}]}\right)$

- Then for $T \to \infty$ followed by $N \to \infty$

$$T\sqrt{N}\widehat{\phi} + 3\sqrt{N} \stackrel{d}{\to} N[0, 10.2]$$

  - need N small relative to T so $\sqrt{N}/T \to 0$
  - convergence rate is $T\sqrt{N}$ rather than usual $\sqrt{TN}$.
    - Like single time series where T not $\sqrt{T}$.
  - centering is wrong: $\widehat{\phi} + 3/T$ is centered on zero ($\widehat{\alpha}_i$ Nickell bias).
    - e.g. $\widehat{\phi} = -0.2$ looks like not unit root, but with $T = 15$, $\widehat{\phi} + 3/T = -0.2 + 3/15 = 0.0$.
  - asymptotic variance of $\widehat{\phi}$ equals $10.2/NT^2$ is not the usual variance
  - lesson: need to recenter and rescale $\widehat{\phi}$.
- If we use the t-statistic $t_{\phi=0} = \widehat{\phi}/se_{\widehat{\phi}}$ (rather than $\widehat{\phi}$) we get

$$\frac{\sqrt{1.25} \times t_{\phi=0} + \sqrt{1.875N}}{\sqrt{645/112}} \stackrel{d}{\to} N[0, 1].$$

  - again need to recenter and rescale.

Summary of Approaches

- 1. Common ρ - one pooled augmented DF regression
  - Detrend data, rescale to common variance for each i, and do ADF
  - Levin and Lin (1992) and Levin, Lin and Chu (2002)
  - Breiting (2000) proposes first pre-whitening data - better power
  - Harris-Tzavalis (1999) have variation that works for T fixed
  - Breiting and Das (2005) extend to weak (spatial) cross-section dependence
- 2. Common ρ - generalize the KPSS test
  - Hadri (2000) LM test
- 3. Heterogeneous ρ_i - N separate ADF t-statistics are averaged
  - detrend data, get ADF t-statistic for each, use the average of these
  - Im, Pesaran and Shin (2003)
  - Pesaran (2006) extends to strong (common shocks) cross-section dependence
- 4. Heterogeneous ρ_i - separate tests and combine p-values
  - Fisher-type tests - Choi (2001) proposes four ways to combine.

Stata Command xtunitroot

Test	Options	Asymptotics	ρ under H_a	Panels
LLC	`noconstant`	$\sqrt{N}/T \to 0$	common	balanced
LLC		$N/T \to 0$	common	balanced
LLC	`trend`	$N/T \to 0$	common	balanced
HT	`noconstant`	$N \to \infty$, T fixed	common	balanced
HT		$N \to \infty$, T fixed	common	balanced
HT	`trend`	$N \to \infty$, T fixed	common	balanced
Breitung	`noconstant`	$(T, N) \to_{\text{seq}} \infty$	common	balanced
Breitung		$(T, N) \to_{\text{seq}} \infty$	common	balanced
Breitung	`trend`	$(T, N) \to_{\text{seq}} \infty$	common	balanced
IPS		$N \to \infty$, T fixed or N and T fixed	panel-specific	unbalanced
IPS	`trend`	$N \to \infty$, T fixed or N and T fixed	panel-specific	unbalanced
IPS	`lags()`	$(T, N) \to_{\text{seq}} \infty$	panel-specific	unbalanced
IPS	`trend lags()`	$(T, N) \to_{\text{seq}} \infty$	panel-specific	unbalanced
Fisher-type		$T \to \infty$, N finite or infinite	panel-specific	unbalanced
Hadri LM		$(T, N) \to_{\text{seq}} \infty$	(not applicable)	balanced
Hadri LM	`trend`	$(T, N) \to_{\text{seq}} \infty$	(not applicable)	balanced

6. Application: Tests follow Stata Manual [XT]

- Eviews 8 has the same tests except for Harris-Tzavalis.
- Generally default option of constant (α_i) no trend $(\delta_i = 0)$ (for `lnxrate`).
  - 1. LLC: Levin-Lin-Chu (1993, 2002)
  - 2. HT: Harris-Tzavalis (1999)
  - 3. Breitung: Breitung (2000)
  - 4. IPS: Im-Pesaran-Shin (2003)
  - 5. IPS: Im-Pesaran-Shin (2003) with serially correlated errors
  - 6. Fisher: Fisher-type Maddala and Wu (1999), Choi (2001)
  - 7. Hadri: LM test (Hadri (2000))
- Test 1 is for small N relative to T so use only G7 $N = 7$
- Test 2 is for large N so all countries N =151 (also use for test 6).
- Tests 3-5, 7 use intermediate N so use OECD $N = 27$
- All tests reject hypothesis that all unit roots. Some are stationary.

Levin-Lin-Chu Panel Unit Root Test

- Levin, Lin and Chu (2002).
- For $N \rightarrow \infty$ slower than $T \rightarrow \infty$ so $N/T \rightarrow 0$
  - though in noconstant case $\sqrt{N}/T \rightarrow 0$.
- ADF model

$$\begin{aligned} \Delta y_{it} &= \phi y_{i,t-1} + \mathbf{z}'_{it}\gamma_i + \textstyle\sum_{j=1}^{p_i} \theta_{ij}\Delta y_{i,t-j} + u_{it} \\ u_{it} &\sim \text{iid}[0, \sigma_i^2] \end{aligned}$$

- Test $H_0 : \phi_i = 0$ for all i (all have unit roots)

  against $H_a : \phi_i = \phi < 0$ for all i (all have common non-unit root).
- Implementation
  - detrend: form panel-by-panel (standardized) regression residuals
    - $\widetilde{e}_{it} = \widehat{e}_{it}/\widehat{\sigma}_{\varepsilon i}$ where $\widehat{e}_{it} = \Delta y_{it} - \sum_{j=1}^{p_i} \widehat{\theta}_{ij}\Delta y_{i,t-j} - \mathbf{z}'_{it}\widehat{\gamma}_i$
    - $\widetilde{v}_{i,t-1} = \widehat{v}_{i,t-1}/\widehat{\sigma}_{\varepsilon i}$ where $\widehat{v}_{i,t-1} = y_{i,t-1} - \sum_{j=1}^{p} \widetilde{\theta}_{ij}\Delta y_{i,t-j} - \mathbf{z}'_{it}\widetilde{\gamma}_i$
  - base test on $\widehat{\delta}/se(\widehat{\delta})$ from $\widetilde{e}_{it} = \widehat{v}_{i,t-1}\delta + error$
  - recenter, rescale and use ratio of short-run to long-run variance.

- Test with lag length determined by AIC rejects H_0 as $p < .05$
  - conclude that stationary
  - if add option `demean` $p = 0.0187$ so again stationary

```
. * Levin-Lin-Chu unit root test
. xtunitroot llc lnrxrate if g7, lags(aic 10)

Levin-Lin-Chu unit-root test for lnrxrate
------------------------------------------
Ho: Panels contain unit roots               Number of panels  =      6
Ha: Panels are stationary                   Number of periods =     34

AR parameter: Common                        Asymptotics: N/T -> 0
Panel means:  Included
Time trend:   Not included

ADF regressions: 1.00 lags average (chosen by AIC)
LR variance:     Bartlett kernel, 10.00 lags average (chosen by LLC)
------------------------------------------------------------------------------
                    Statistic      p-value
------------------------------------------------------------------------------
 Unadjusted t        -6.7538
 Adjusted t*         -4.0277        0.0000
------------------------------------------------------------------------------
```

- Previous chose only one lag ($p_i = 1$ on average) - seems too short
  - setting four lags of Δy_{it} we do not reject H_0
  - conclude that unit root present.

```
. * So do unit root tests with lag lengths set to 4
. xtunitroot llc lnrxrate if g7, lags(4)

Levin-Lin-Chu unit-root test for lnrxrate
-----------------------------------------
Ho: Panels contain unit roots               Number of panels  =      6
Ha: Panels are stationary                   Number of periods =     34

AR parameter: Common                        Asymptotics: N/T -> 0
Panel means:  Included
Time trend:   Not included

ADF regressions: 4 lags
LR variance:     Bartlett kernel, 10.00 lags average (chosen by LLC)
------------------------------------------------------------------------------
                    Statistic      p-value
------------------------------------------------------------------------------
 Unadjusted t         -4.9507
 Adjusted t*          -1.1601        0.1230
------------------------------------------------------------------------------
```

Harris-Tsavalis Panel Unit Root Test

- Harris and Tsavalis (1999).
- For T fixed and $N \rightarrow \infty$

$$\begin{aligned} y_{it} &= \rho y_{i,t-1} + \mathbf{z}_{it}'\gamma_i + u_{it} \\ u_{it} &\sim N[0, \sigma^2] \end{aligned}$$

  - **very strong assumptions** - i.i.d. error ! (DF not ADF)
  - can relax normality (but then need to estimate kurtosis of u_{it})
  - and can include time dummies
- Test $H_0 : \phi_i = 0$ for all i (all have unit roots)

  against $H_a : \phi_i = \phi < 0$ for all i (all have common non-unit root).
- $\widehat{\rho}$ biased for finite T via γ_i
  - HT find the bias correction term when $\rho = 1$
  - e.g. if $\mathbf{z}_{it}'\gamma_i = \alpha_i$ then

    $\sqrt{N}(\widehat{\rho} - (1 - \frac{3}{T+1})) \xrightarrow{d} N\left[0, \frac{3(17T^2-20T+17)}{5(T-1)(T+1)^3}\right]$.

- Reject H_0 as $p < .05$. Conclude that stationary.

```
. * Harris-Tsavalis unit root test
. xtunitroot ht lnrxrate, demean

Harris-Tzavalis unit-root test for lnrxrate
-------------------------------------------
Ho: Panels contain unit roots              Number of panels  =    151
Ha: Panels are stationary                  Number of periods =     34

AR parameter: Common                       Asymptotics: N -> Infinity
Panel means:  Included                                  T Fixed
Time trend:   Not included                 Cross-sectional means removed
------------------------------------------------------------------------------
                    Statistic        z        p-value
------------------------------------------------------------------------------
 rho                  0.8184     -13.1239     0.0000
------------------------------------------------------------------------------
```

Breitung Panel Unit Root Test

- Breitung (2000) and Breitung and Das (2005)
- For $T \rightarrow \infty$ followed by $N \rightarrow \infty$
- Have heterogeneous intercept, common trend

$$y_{it} = \rho y_{i,t-1} + \gamma_i + \beta t + u_{it};\ \ u_{it} \sim \text{serially correlated and not iid}$$

- LLC controls for heterogeneous γ_i by mean differencing
  - instead use the long difference $y_{it} - y_{i,p+1}$
    - with p lags lost due to serial correlation in u_{it}
  - then $y_{it} = \gamma_i + \beta t$ and $y_{i,p+1} = \gamma_i$ so $y_{it} - y_{i,p+1} = \beta(t-p+1)$.
- To control for serial correlation of order p in u_{it} pre-whiten data
  - let $\Delta\widehat{\varepsilon}_{it}$ be residual from OLS of Δy_{it} on $\Delta y_{i,t-1}, ..., \Delta y_{i,t-p}$
  - let $\widehat{\varepsilon}^{*}_{it}$ be residual from OLS of $y^{*}_{it} = y_{it} - y_{i,p+1}$ on $\Delta y_{i,t-1}, ..., \Delta y_{i,t-p}$
  - $\widehat{\lambda} = \frac{\sum_{i=1}^{N}\sum_{t=p+2}^{T}\widehat{\varepsilon}^{*}_{it}\Delta\widehat{\varepsilon}_{it}/\widehat{\sigma}^2_i}{\sqrt{\sum_{i=1}^{N}\sum_{t=p+2}^{T}\Delta\widehat{\varepsilon}^2_{it}/\widehat{\sigma}^2_i}} \xrightarrow{d} N\left[0,1\right]$ where $\sigma^2_i = \frac{1}{T-p-2}\sum_{t=p+2}^{T}\Delta\widehat{\varepsilon}^2_{it}$.
- A variation (robust) permits weak cross-sectional correlation
  - though then need $T >> N$.

- This version has no lags (should add lags) and gets cross-section correlation robust version
  - marginally reject H_0 at 5%.

```
. * Breitung unit root test
. xtunitroot breitung lnrxrate if oecd, robust

Breitung unit-root test for lnrxrate
--------------------------------------
Ho: Panels contain unit roots            Number of panels  =     27
Ha: Panels are stationary                Number of periods =     34

AR parameter: Common                     Asymptotics:  T,N -> Infinity
Panel means:  Included                                    sequentially
Time trend:   Not included               Prewhitening: Not performed
------------------------------------------------------------------------------
                    Statistic      p-value
------------------------------------------------------------------------------
 lambda*             -1.6794        0.0465
------------------------------------------------------------------------------
 * Lambda robust to cross-sectional correlation
```

Im-Pesaran-Shin Panel Unit Root Test

- Im, Pesaran and Shin (2003) allow different ϕ_i for each panel
- For T fixed (need normal errors) or $T \rightarrow \infty$ followed by $N \rightarrow \infty$.
- Augmented DF model

$$\begin{aligned} \Delta y_{it} &= \phi_i y_{i,t-1} + \mathbf{z}_{it}'\gamma_i + \sum\nolimits_{j=1}^{p_i} \theta_{ij} \Delta y_{i,t-j} + u_{it} \\ u_{it} &\sim N[0, \sigma_i^2] \text{ in finite } T \text{ case} \end{aligned}$$

- Test $H_0 : \phi_i = 0$ for all i (all have unit roots)

  against $H_a : \phi_i = \phi < 0$ for some i (some have non-unit root).

- Obtain t statistics for each panel
  - these are usual augmented Dickey-Fuller t-test denote τ_i
  - then τ_i under $H_0 : \phi_i = 0$ has mean $\mathrm{E}[\tau_i]$ and variance $\mathrm{V}[\tau_i]$ that are given in tables
  - the average of these recentered and rescaled $\xrightarrow{d} N[0,1]$ by CLT as $N \rightarrow \infty$
    - unbalanced panel
      $Z = \sqrt{N}\left[\frac{1}{N}\sum_{i=1}^{N}\tau_i - \frac{1}{N}\sum_{i=1}^{N}\mathrm{E}[\tau_i]\right] / \sqrt{\frac{1}{N}\sum_{i=1}^{N}\mathrm{V}[\tau_i]} \xrightarrow{d} N[0,1]$
    - balanced panel $Z = \sqrt{N}\left[\frac{1}{N}\sum_{i=1}^{N}\tau_i - \mathrm{E}[\tau]\right] / \sqrt{\mathrm{V}[\tau_i]} \xrightarrow{d} N[0,1]$

- With serially uncorrelated errors (no lags)
  - reject as $p < 0$ and conclude some panels are stationary

```
. * Im-Pesaran-Shin unit root test
. xtunitroot ips lnrxrate if oecd, demean

Im-Pesaran-Shin unit-root test for lnrxrate
--------------------------------------------
Ho: All panels contain unit roots             Number of panels  =     27
Ha: Some panels are stationary                Number of periods =     34

AR parameter: Panel-specific                  Asymptotics: T,N -> Infinity
Panel means:  Included                                          sequentially
Time trend:   Not included                    Cross-sectional means removed

ADF regressions: No lags included
------------------------------------------------------------------------------
                                               Fixed-N exact critical values
                    Statistic      p-value          1%       5%      10%
------------------------------------------------------------------------------
 t-bar               -3.1327                     -1.810   -1.730   -1.680
 t-tilde-bar         -2.5771
 Z-t-tilde-bar       -7.3911        0.0000
------------------------------------------------------------------------------
```

- With serially correlated errors (AIC has average 1.5 lags)
  - reject as $p < 0$ and conclude some panels are stationary

```
. * Im-Pesaran-Shin unit root test with serially correlated errors
. xtunitroot ips lnrxrate if oecd, lags(aic 8) demean

Im-Pesaran-Shin unit-root test for lnrxrate
-------------------------------------------
Ho: All panels contain unit roots           Number of panels  =     27
Ha: Some panels are stationary              Number of periods =     34

AR parameter: Panel-specific                Asymptotics: T,N -> Infinity
Panel means:  Included                                       sequentially
Time trend:   Not included                  Cross-sectional means removed

ADF regressions: 1.48 lags average (chosen by AIC)
------------------------------------------------------------------------------
                    Statistic      p-value
------------------------------------------------------------------------------
 W-t-bar             -7.3075        0.0000
------------------------------------------------------------------------------
```

Fisher-Type Panel Unit Root Test

- Maddala and Wu (1999) and Choi (2001).
- For $T \longrightarrow \infty$ and N fixed
  - Model is completely heterogeneous
- Do separate time series unit root tests in each panel
  - Stata `fisher` does Dickey-Fuller and Phillips-Perron
- Then need to combine the N tests.
- This is like meta-analysis
  - combine the p values using methods of Fisher
  - e.g. $P = -2\sum_{i=1}^{N} \ln p_i \sim \chi^2(2N)$ (reject H_0 if large)
  - there are other ways of combining - see the output.

- Here do Dickey-Fuller with two lags in each panel
  - all four ways have $p <$ .05 so conclude at least one panel is stationary.

```
. * Fisher unit root test
. xtunitroot fisher lnrxrate, dfuller drift lags(2) demean

Fisher-type unit-root test for lnrxrate
Based on augmented Dickey-Fuller tests
---------------------------------------
Ho: All panels contain unit roots           Number of panels  =    151
Ha: At least one panel is stationary        Number of periods =     34

AR parameter: Panel-specific                Asymptotics: T -> Infinity
Panel means:  Included
Time trend:   Not included                  Cross-sectional means removed
Drift term:   Included                      ADF regressions: 2 lags
------------------------------------------------------------------------------
                                  Statistic      p-value
------------------------------------------------------------------------------
 Inverse chi-squared(302)   P      975.9130       0.0000
 Inverse normal             Z      -19.6183       0.0000
 Inverse logit t(759)       L*     -20.9768       0.0000
 Modified inv. chi-squared Pm       27.4211       0.0000
------------------------------------------------------------------------------
 P statistic requires number of panels to be finite.
 Other statistics are suitable for finite or infinite number of panels.
------------------------------------------------------------------------------
```

Hadri LM Panel Unit Root Test

- Hadri (2000) generalizes the KPSS test.
- $T \rightarrow \infty$ and then $N \rightarrow \infty$
- Suppose

$$\begin{aligned} y_{it} &= r_{it} + \beta_i t + \varepsilon_{it}, & \varepsilon_{it} \sim N[0, \sigma_\varepsilon^2] \\ r_{it} &= r_{i,t-1} + u_{it}, & u_{it} \sim N[0, \sigma_u^2]. \end{aligned}$$

- In general y_{it} will be I(1) due to the random walk r_{it}.
- But if $\sigma_u^2 = 0$ then $u_{it} = 0$, $r_{it} = r_{i,t-1} =$ constant
  - so $y_{it} = \alpha + \beta_i t + \varepsilon_{it}$ is trend stationary.
- So test $H_0 : \sigma_u^2 / \sigma_\varepsilon^2 = 0$ against $H_a : \sigma_u^2 / \sigma_\varepsilon^2 = 0$
  - Note that now H_0 is no unit root against H_q unit root
  - simulations show this tends to over-reject.
- The test is an LM test that $\overset{d}{\rightarrow} N[0, 1]$ after recentering and rescaling
  - OLS regress $y_{it} = \mathbf{d}_{it}'\gamma + u_t$, $\widehat{u}_{it} = y_{it} - \mathbf{d}_{it}'\widehat{\gamma}$, $S_{it} = \sum_{s=1}^{t} \widehat{u}_{is}^2$
  - $LM = \sum_{i=1}^{N} \left(\sum_{t=1}^{T} S_{it}^2 / (T^2 f_{i0}) \right)$; $f_{i0} = \widehat{\mathrm{V}}_{long-run}[\widehat{u}_{it}]$.

- Reject H_0 as $p < .05$ here means some panels contain unit roots.

```
. * Hadri unit root test
. xtunitroot hadri lnrxrate if oecd, kernel(bartlett 5) demean

Hadri LM test for lnrxrate
----------------------------
Ho: All panels are stationary              Number of panels  =     27
Ha: Some panels contain unit roots         Number of periods =     34

Time trend:         Not included           Asymptotics: T, N -> Infinity
Heteroskedasticity: Robust                                  sequentially
LR variance:        Bartlett kernel, 5 lags Cross-sectional means removed
------------------------------------------------------------------------------
                    Statistic       p-value
------------------------------------------------------------------------------
 z                     9.6473        0.0000
------------------------------------------------------------------------------
```

7. Cross-section Dependence

- Levin, Lin and Chu (2003, p.13) state that cross-sectional dependence can be partially controlled for on their tests by first subtracting the "cross-sectional average" $\bar{y}_t = \frac{1}{N}\sum_{i=1}^{N} y_{it}$
  - equivalent to including a full set of time dummies in the original model for y_{it}
  - this is a single common shock that has an identical effect on all individuals in the panel.
  - Stata gives this as an option for all its unit tests, citing LLC
  - but LLC only say that this is okay for their test.
- Breiting and Pesaran (2008) call tests that allow for cross-section correlation of errors "second generation tests".
  - ignoring cross-section dependence leads to size bias
  - especially if the panel units are cross-cointegrated e.g. PPP

Weak and Strong Cross-Section Dependence

- Weak dependence
  - eigenvalues of covariance matrix of y_{it} are bounded as $N \longrightarrow \infty$
  - essentially this means that correlations across individuals need to dampen as they get "further" apart
  - this is the case for spatial dependence
- Strong dependence
  - at least one eigenvalue of covariance matrix of y_{it} diverges as $N \longrightarrow \infty$
  - this is the case for common factors $\mathbf{F}_i' \delta_t$.

Unit Root Tests

- Breitung and Pesaran (2008) section 9.4 have summary
  - see this for references
- Moon and Perron (2004) and Pesaran (2007)
  - model $y_{it} = (1 - \rho_i)\mu_i + \rho_i y_{i,t-1} + u_{it}$, $u_{it} = \gamma_i f_t + \varepsilon_{it}$
  - test of $\pi_i = 1$ is joint test that (1) all time series I(1); and (2) they are not cointegrated.
- Bai and Ng (2004) propose Panel Analysis of Nonstationarity in Idiosyncratic and Common components (PANIC)
  - analyzes common factors and idiosyncratic components separately
  - nonstationarity can be pervasive, variable-specific or both
  - can determine the number of independent stochastic trends driving the common factors.

Pesaran (2007)

- Cross-sectionally augmented Dickey Fuller (CADF) test.
- One-factor model $u_{it} = \gamma_i f_t + \varepsilon_{it}$ for the cross-panel correlation.
- Begin with no serial correlation

$$\begin{array}{rl} & y_{it} = (1-\rho_i)\mu_i + \rho_i y_{i,t-1} + u_{it}; \quad u_{it} = \gamma_i f_t + \varepsilon_{it} \\ \Longrightarrow & \Delta y_{it} = \alpha_i + \phi_i y_{i,t-1} + \gamma_i f_t + \varepsilon_{it};\ \alpha_i = \mu_i(1-\rho_i);\ \phi_i = (\rho_i - 1) \end{array}$$

  - ε_{it} iid $[0, \sigma_i^2]$, f_t iid $[0, \sigma_f^2 = 1]$, ε_{it}, f_t and γ_i independent for all i
- Test $H_0 : \phi_i = 0$ for all i against $H_a : \phi_i < 0$ for some i.
- Pesaran (2006) shows f_t can be proxied by cross-section mean $\bar{y}_t = \frac{1}{N}\sum_{i=1}^N y_{it}$ and lags if $N \rightarrow \infty$
  - So OLS estimate

$$\Delta y_{it} = a_i + b_i y_{i,t-1} + c_i \bar{y}_{t-1} + d_i \Delta\bar{y}_t + e_{it}$$

  - If T is fixed replace y_{it} with $y_{it} - \bar{y}_0$; $\bar{y}_0 = \frac{1}{N}\sum_{i=1}^N y_{i0}$.
- $CADF_i$ Test for panel i: $t_i = \widehat{b}_i / se[\widehat{b}_i]$.
- $\overline{CADF}$ Joint panel test: $\frac{1}{N}\sum_{i=1}^N t_i$ where truncate as in his equation (34)

- Pesaran (2007) with serially correlated error (AR(p)).
- OLS estimate (his equation (54))

$$\Delta y_{it} = a_i + b_i y_{i,t-1} + c_i \bar{y}_{t-1} + \sum_{j=0}^{p} d_{ij} \Delta \bar{y}_{t-j} + \sum_{j=1}^{p} \delta_{ij} \Delta \bar{y}_{i,t-j} + e_{it}$$

- CADF_i Test for panel i: $t_i = \widehat{b}_i / se[\widehat{b}_i]$
  - has critical values in Tables 1.
- $\overline{\text{CADF}}$ Joint panel test: $\frac{1}{T} \sum_{i=1}^{N} t_t$ where truncate as in his equation (34)
  - has critical values in Tables 2
  - or can combine CADF_i test p-values e.g. $-2 \sum_{i=1}^{N} \ln(p_i) \sim \chi^2(2N)$.
- The critical values are obtained by simulation (no $N[0, 1]$ tests here)
  - asymptotic theory generally the same under sequential or joint limits
  - $N = 10, 20, 30, 40, 50, 100$ and $T = 10, 20, 30, 40, 50, 100$
  - generally works well even for small N and T.
- The paper also has good summary of other methods in the introduction and implements some of these in simulation and application.

Application: Pesaran (2007)

```
use pennxrate.dta, clear
rename lnrxrate yy
keep if oecd==1
bysort year: egen ytbar = mean(yy)
sort id year
statsby phi=_b[l.yy] se=_se[l.yy], by(id) clear: regress d.yy l.yy l.ytbar L(0/2)d.ytbar
L(1/2)d.yy
generate t = phi/se
* t is CADF_i test: Critical values are in Table 1 (individual panel)
format phi t %9.3f
list phi t, clean
* Panel ACDF test: Critical values are in Table 2 (average of panels)
mean t
* Fisher test combines the individual test p values
generate lnp = ln(p)
quietly summarize lnp
display "Fisher test = " -2*r(sum) " with p-value = "chi2tail(r(N),-2*r(sum))
```

Application (continued)

- Individual CADF_i t-tests
  - critical value -3.34 for $N = 30$ and $T = 30$ from Table 1(b)
  - here most negative t's are -4.18, -4.09, -3.-04
  - so 2 of 27 are rejected - borderline reject H_0
- Overall $\overline{\text{CADF}}$ test
  - critical value -2.15 for $N = 30$ and $T = 30$ from Table 2(b)
  - here `mean t` = -1.78
  - do not reject H_0 : all $\phi_i = 0$ against H_a : some $\phi_i \neq 0$.
- Fisher test combining individual CADF_i t-tests
  - Fisher test = 8.6578493 with p-value = .00325655
  - reject H_0 : all $\phi_i = 0$ against H_a : some $\phi_i \neq 0$.

8. Random Walk is Asymptotically a Wiener Processes

- Let $y_t = y_{t-1} + u_t = \sum_{s=1}^{t} u_s$ where $u_s \sim$ iid $N[0,1]$
  - then $y_t \sim N[0,t]$
- Change the index from $0 \leq t \leq T$ to $0 \leq r \leq 1$
  - $t = [rT]$ where $[rT]$ is the integer part of rT
  - $y_t = y_{[rT]} \sim N[0,t]$
  - $\frac{1}{\sqrt{T}} y_{[rT]} \sim N[0,[rT]/T] \quad \Longrightarrow \quad \frac{1}{\sqrt{T}} y_{[rT]} \sim N[0,r]$
  - $\frac{1}{\sqrt{T}} y_{[rT]} - \frac{1}{\sqrt{T}} y_{[r'T]} \sim N[0,r-r']$; $r' < r$ similarly
- But this is just a discrete time version of a continuous time Wiener process (Brownian motion) $W(r)$, defined as
  - $W(r) = 0$
  - $W(r)$ is continuous in r almost surely, $0 \leq r \leq 1$
  - $W(r) - W(r') \sim N[0,r-r']$ independently for any $0 \leq r' < r \leq 1$
- Key Result: $\frac{1}{\sqrt{T}} y_{[rT]} \overset{a}{=} W(r)$ or $\frac{1}{\sqrt{T}} y_t \overset{a}{=} W(\frac{t}{T})$

DF Test involves Sums of Functions of Random Walks

- Dickey-Fuller test from regress Δy_t on y_{t-1} yields
  - $\widehat{\phi} = \widehat{\rho} - 1 = \sum_{t=1}^{T} y_{t-1}(\Delta y_t) / \sum_{t=1}^{T} y_{t-1}^2$
- For nonrandom function $f(r)$ defined on $[0,1]$
  - $\int_0^1 f(r)dr = \lim_{T\to\infty} \sum_{t=1}^{T} f(\frac{t}{T}) \times \frac{1}{T} = \lim_{T\to\infty} \frac{1}{T} \sum_{t=1}^{T} f(\frac{t}{T})$
  - this extends to random functions such as $W(r)$
- So for sums of y_t and its products, given $\frac{1}{\sqrt{T}} y_t \overset{a}{=} W(\frac{t}{T})$
  - $\frac{1}{T^{3/2}} \sum_{t=1}^{T} y_t = \frac{1}{T} \sum_{t=1}^{T} \frac{1}{\sqrt{T}} y_t \overset{a}{=} \frac{1}{T} \sum_{t=1}^{T} W(\frac{t}{T}) \overset{d}{\to} \int_0^1 W(r)dr$
  - $\frac{1}{T^2} \sum_{t=1}^{T} y_t^2 = \frac{1}{T} \sum_{t=1}^{T} (\frac{1}{\sqrt{T}} y_t)^2 = \frac{1}{T} \sum_{t=1}^{T} W(\frac{t}{T})^2 \overset{d}{\to} \int_0^1 W(r)^2 dr$
  - $\frac{1}{T} \sum_{t=1}^{T} y_{t-1} \Delta y_t \overset{d}{\to} \frac{1}{2}[W(1)^2 - 1]$ after some algebra
- So Dickey-Fuller z-test statistic
  - $T\widehat{\phi} = \frac{\frac{1}{T} \sum_{t=1}^{T} y_{t-1}(\Delta y_t)}{\frac{1}{T^2} \sum_{t=1}^{T} y_{t-1}^2} \overset{d}{\to} \frac{\frac{1}{2}[W(1)^2 - 1]}{\int_0^1 W(r)^2 dr}$ (no nuisance parameters)

Generalizations

- u_t in $y_t = y_{t-1} + u_t$ can be nonnormal, serially correlated, heterogeneously distributed if $T \rightarrow \infty$
  - we need $\frac{1}{\sqrt{T}} y_{[rT]} = \frac{1}{\sqrt{T}} \sum_{s=1}^{t} u_s \xrightarrow{d} \omega \times N[0, r]$ for some ω
  - this can be established using a (functional) central limit theorem
  - ω is the long-run variance of $y_{[rT]}$
- We often include an intercept and possibly trend in the DF test
  - $\Delta y_t = \alpha + \delta t + \phi y_{t-1} + u_t$
  - then need results for $\sum_{t=1}^{T} \Delta y_t$, $\sum_{t=1}^{T} t\Delta y_t$, $\sum_{t=1}^{T} y_{t-1}$, $\sum_{t=1}^{T} t y_{t-1}$, ... see Hamilton (1996, p.506)
  - or can use results for demeaned processes (Frisch-Waugh) ... see Hayashi (2000, p.570)

- Consider $\Delta y_t = \alpha + \phi y_{t-1} + u_t$
  - use demeaned random walk $y_t^{\mu} = y_t - \bar{y}$; where $\bar{y} = \frac{1}{T}\sum_{t=1}^{T} y_t$
  - continuous time analog is detrended standard Wiener $W^{\mu}(r) = W(r) - \int_0^1 W(s)ds$
  - $\frac{1}{T^2}\sum_{t=1}^{T}(y_t^{\mu})^2 \xrightarrow{d} \int_0^1 W^{\mu}(r)^2 dr$
  - $\frac{1}{T}\sum_{t=1}^{T} y_{t-1}^{\mu}\Delta y_t^{\mu} \xrightarrow{d} \frac{1}{2}[W^{\mu}(1)^2 - W^{\mu}(0)^2 - 1]$
- Consider $\Delta y_t = \alpha + \delta + \phi y_{t-1} + u_t$ detrended random walk $y_t^{\mu} = y_t - \widehat{\alpha} - \widehat{\delta} t$ where $\widehat{\alpha}$ and $\widehat{\delta}$ are from OLS of y_t on intercept and linear time trend
  - then different functions of Wiener processes.

9. Summary

- Stata unit root test

Single Time Series	`dfgls, dfuller`
Panel: cross-country correlation	`xtunitroot`
Panel: cross-country correlation	—

- For panel unit root tests with cross-country correlation can easily code up Pesaran (2007) CADF test and refer to tables in his paper.

10. Selected References

- Surveys

  - Breitung, J, and M.H. Pesaran (2008), "Unit Roots and Cointegration in Panels," in L. Mátyás and P. Sivestre, eds., The Econometrics of Panel Data: Advanced Studies in Theoretical and Applied Econometrics, Third Edtiion, Advanced Studies in Theoretical and Applied Econometrics, 46, 2008, pp 279-322, Springer.
  - Kirchgässner. G. (2013), Introduction to Modern Time Series Analysis, chapter 7, Springer.
  - Stata Manual [XT] Longitudinal Data / Panel Data entry xtunitroot Method and Formulas summarizes several panel unit root tests.
  - Eviews 8 Users Guide II Chapter 16 summarizes several panel unit root tests.

- Unit Root Tests without cross-section correlation

  - Im, K.S., M.H. Pesaran, and Y. Shin (2003), "Testing for Unit Roots in Heterogenous Panels," Journal of Econometrics, 115, 53–74.
  - Levin, A., and C. Lin, and C.J. Chu (1993), "Unit Root Tests in Panel Data: Asymptotic and Finite Sample Properties," Journal of Econometrics," UCSD Economics Discussion Paper 92-23.
  - Levin, A., C. Lin, and C.J. Chu (2002), "Unit Root Tests in Panel Data: Asymptotic and Finite Sample Properties," Journal of Econometrics," 108, 1–24.

- Unit Root Tests with cross-section correlation

  - Breitung, J. and S. Das (2005), "Panel Unit Root Tests Under Cross Sectional Dependence," Statistica Neerlandica, 59, 414–433.
  - Pesaran, M.H. (2007), "A Simple Panel Unit Root Test in the Presence of Cross Section Dependence, Journal of Applied Econometrics," 22, 265–312.
  - Bai, J. and S. Ng (2004), "A PANIC Attack on Unit Roots and Cointegration," Econometrica, 72, 1127-1177.